

## EXCEL 2003 « LES FONCTIONS »

### I. INTRODUCTION :

La fonction première d'Excel est de faire des calculs en utilisant les valeurs entrées dans un tableau. Ces calculs peuvent être de simples additions ou des calculs plus compliqués tels que la valeur future d'un placement à terme.

Une formule est une donnée dans une cellule du chiffrier qui a pour tâche d'effectuer un calcul quelconque. Elle est composée de valeurs, d'opérateurs mathématiques, d'adresses de cellules, de noms de champs, d'autres formules, de fonctions intégrées ou même, dans certains cas, de texte.

### ADDITIONS, SOUSTRATIONS, MULTIPLICATIONS ET DIVISIONS

Certains des opérateurs que nous utiliserons avec Excel diffèrent des opérateurs que beaucoup d'entre nous avons appris. Voici les opérateurs qu'il faudra utiliser pour faire les calculs suivants :

Calculs	Opérateurs
Addition	+
Soustraction	-
Multiplication	*
Division	/

#### I.1. La formule :

Une formule doit toujours débuter par un signe (+, -, ou =) suivi de l'équation.

##### Exemple 01:

+A1+A2  
=somme(A1:Z20)  
=moyenne(A1:Z20)  
=max(A1 :A4)

Cette équation est constituée d'une adresse (exemple : **A1** : colonne **A**, ligne **1**) ou d'un nombre fixe (exemple : 100). Il est possible d'utiliser la valeur inscrite dans une cellule pour effectuer un calcul, mais il est recommandé d'utiliser les adresses des valeurs dans les formules. Il existe une bonne raison à cela. Si, un jour, on modifie la valeur d'une cellule, le résultat du calcul sera réajusté en prenant en considération le changement que nous venons de faire. Si nous avons utilisé la valeur au lieu de l'adresse dans la formule, nous devons la modifier à chaque fois que l'on change le contenu d'une cellule.

### Exemple 02:

Si nous voulons additionner la valeur que nous avons dans **A1** à celle qui se trouve dans **B1**, la formule serait la suivante : **=A1+B1**. Si on oublie d'insérer le signe d'égalité, Excel considérera cette entrée comme une donnée texte et non une formule. Le texte sera inscrit dans la cellule et non le résultat du calcul. N'oubliez pas de toujours déplacer votre curseur dans la cellule où le résultat doit être affiché.

## II. LES FONCTIONS ARITHMETIQUES :


Les différentes opérations arithmétiques possibles vous sont proposées sous forme d'exemples.

### 1. ADDITION :

#### Exemple :


Allez sur le menu supérieur **Fichier** et cliquez sur **Nouveau** : un classeur apparaîtra. Remplissez les cellules **A1** à **A4** avec les données (Info 1, Info 2, ...) et les cellules **B1** à **B4** avec les données (10, 20, ...).

Dans la cellule **B5**, entrez la formule suivante : **=B1+B2+B3+B4** et tapez sur la touche **Entrée** du clavier pour avoir la réponse.



	A	B
1	INFO 1	10
2	INFO 2	20
3	INFO 3	10
4	INFO 4	40
5		=B1+B2+B3+B4
6		

On obtient le résultat suivant :



	A	B
1	INFO 1	10
2	INFO 2	20
3	INFO 3	10
4	INFO 4	40
5		80
6		

### 2. SOUSTRACTION :

**Exemple :** Utiliser le même exemple que l'addition, mais cette fois-ci entrez la formule suivante : **=B1-B2-B3-B4** dans la cellule **B5**, et tapez sur la touche **Entrée** du clavier pour avoir la réponse.

**REMARQUE :** Utiliser le même principe pour la multiplication et la division.

### III. LES FONCTIONS PREPROGRAMMEES :

Les fonctions préprogrammées sont des formules Excel qui permettent de calculer automatiquement le maximum, le minimum, la moyenne et le nombre de données contenues dans un classeur.

Les différentes fonctions possibles vous sont proposées sous forme d'exemples.

#### 1. La fonction maximum :

Cette fonction renvoie le nombre le plus grand d'une plage de cellules donnée.

##### Exemple :

Allez dans le menu supérieur Fichier et cliquez sur Nouveau : un classeur apparaît. Remplissez les cellules **A1** à **A4** avec les données (info 1, info 2, ...) et les cellules **B1** à **B4** avec les données (10, 20, ...).

Dans la cellule **B5**, entrez la formule = **Max**. Ouvrir une parenthèse puis sélectionnez la cellule B1 avec la souris jusqu'à la cellule B4, fermer la parenthèse et tapez sur la touche Entrée du clavier pour avoir la réponse.

	A	B
1	INFO 1	10
2	INFO 2	20
3	INFO 3	10
4	INFO 4	40
5		=max(B1:B4)
6		

On obtient le résultat 120 qui est le nombre maximum entre l'info 1 et l'info 4.

	A	B
1	INFO 1	10
2	INFO 2	20
3	INFO 3	10
4	INFO 4	40
5		40
6		

## 2. La fonction minimum :

Cette fonction renvoie le nombre le plus petit d'une plage de cellules donnée.

### Exemple :

Allez dans le menu supérieur Fichier et cliquez sur Nouveau : un classeur apparaît. Remplissez les cellules **A1** à **A4** avec les données (info 1, info 2, ...) et les cellules **B1** à **B4** avec les données (10, 20, ...).

Dans la cellule **B5**, entrez la formule = **Min**.

Ouvrir une parenthèse puis Sélectionnez la cellule **B1** avec la souris jusqu'à la cellule **B4**, fermer la parenthèse et tapez sur la touche Entrée du clavier pour avoir la réponse..

Classeur1		
	A	B
1	Info 1	10
2	Info 2	20
3	Info 3	51
4	Info 4	120
5		=min(B1:B4)

On obtient le résultat 10 qui est le nombre entre l'info 1 et l'info 4.

Classeur1		
	A	B
1	Info 1	10
2	Info 2	20
3	Info 3	51
4	Info 4	120
5	Résultat	10

minimum

## 3. La fonction moyenne

Cette fonction calcule la moyenne de plusieurs plages de cellules.

### Exemple :

Allez dans le menu supérieur Fichier et cliquez sur Nouveau : un classeur apparaît. Remplissez les cellules **A1** à **A4** avec les données (info 1, info 2, ...) et les cellules **B1** à **B4** avec les données (10, 20, ...).

Dans la cellule **B5**, entrez la formule = **moyenne**.

Ouvrir une parenthèse puis Sélectionnez la cellule **B1** avec la souris jusqu'à la cellule **B4**, fermer la parenthèse et tapez sur la touche Entrée du clavier pour avoir la réponse..

Microsoft Excel - Classeur1		
	A	B
1	INFO1	10
2	INFO2	20
3	INFO3	10
4	INFO4	10
5		=moyenne(B1:B4)
6		


On obtient le résultat 12,5 qui est la moyenne des données pour les infos 1 à 4.

Microsoft Excel - Classeur1		
	A	B
1	INFO1	10
2	INFO2	20
3	INFO3	10
4	INFO4	10
5		12,5
6		

#### IV. CALCULS AUTOMATIQUES :

EXCEL contient un certain nombre de fonctions prédéfinies.

Il suffit de cliquer sur le bouton fonction (c'est  $\Sigma$  la somme par défaut) et de choisir la fonction que l'on veut utiliser : **Somme, moyenne, Compteur, Max, Min.**




#### Exemple pour la fonction SOMME :

- Sélectionnez les cellules contenant les éléments à additionner.
- Cliquez sur le bouton Somme.

Si la fonction que vous voulez utiliser n'est pas dans cette liste, cliquez sur Autres fonctions.

Lorsque l'on utilise des fonctions prédéfinies, il faut respecter la syntaxe de la fonction sous peine d'obtenir un message d'erreur. Comme il existe environ 400 fonctions prédéfinies et qu'on ne peut mémoriser toutes les syntaxes, EXCEL met à notre disposition un assistant Fonction. Cet outil a pour mission de nous venir en aide pour la saisie des fonctions.



Choisissez une catégorie pour la fonction et la liste apparaîtra en dessous.

## V. LES FORMULES DE CALCULS

### IV.1. Les formules de calculs simples

#### 1. Les formules arithmétiques :

Elles comportent des constantes, des références cellulaires ou des opérateurs arithmétiques. Ce type de formule permet de calculer un résultat et de l'afficher dans la cellule dans laquelle la formule a été saisie.

**Exemples :** =D7\*B17

=-A25

=1500\*B17

Tableau des différents opérateurs :

Opérateur	Exemple	Signification
<b>Opérateurs arithmétiques</b>		
*	10*8	Multiplication
/	B10/B15	Division
-	D17-B17	Soustraction
^	5^5	Puissance
+	6+7	Addition
%	5,5%	Pourcentage
-	-B17	Signe préfixé
<b>Opérateurs de comparaison</b>		
=	A1=A8	Égal
<>	A1<>A8	Différent
<=	A1<=A8	Inférieur ou égal
>=	A1>=A8	Supérieur ou égal
<b>Opérateurs de plage de cellules</b>		
:	A10:D10	Plage de cellules comprises entre A10 et D10
Espace	A10:D10 D10:D15	
;	A10:D10;A15:D10	
<b>Opérateurs de concaténation</b>		
&	A10&D10	Concaténation de chaînes de caractères

#### 2. Les formules de chaînes de caractères :

Elles permettent de combiner deux ou plusieurs chaînes de caractères en utilisant l'opérateur &.

**Exemples :** ="LAURENT"&A20

Il va écrire dans la case sélectionnée le mot LAURENT combiné avec le contenu de la case A20

#### 3. Les formules logiques :

Elles contiennent des comparaisons entre des constantes ou des cellules.

**Exemples :** =A10>A20

=A10>=20

Ce type de formule renvoi la valeur **Vrai** si c'est exact ou **Faux** dans le cas contraire.

## V.2. Les formules de calcul avec des fonctions :

**Exemples** : =SOMME(A10:A20)  
=MAX(A10:A20)

### Tableaux des fonctions les plus courantes

Nom des fonctions	Définition utilité
=SOMME(Plage)	Effectue la somme des cellules comprise dans la <i>plage</i> Ex : =SOMME(A1:A4)
=MOYENNE(Plage)	Effectue la somme des cellules comprise dans la <i>plage</i> Ex : =MOYENNE(A1:A4)
=MIN(Plage)	Détermine la valeur minimale contenue dans la <i>plage</i> Ex : =MIN(A1:A4)
=MAX(Plage)	Détermine la valeur maximale contenue dans la <i>plage</i> Ex : =MAX(A1:A4)
=NBVAL(Plage)	Compte le nombre de cellules non vides contenue dans la <i>Plage</i> Ex : =NBVAL(A1:A4)
=NB(Plage)	Compte le nombre de cellules numériques contenues dans la <i>Plage</i> Ex : =NB(A1:A4)
=ARRONDI(Valeur;Nb décimales)	Arrondit la <i>valeur</i> au plus proche en fonction du <i>nombre de décimales</i> spécifié. Ex. : =ARRONDI(6,55957;2)=6,56 =ARRONDI(6,55957;0)=7  Pour arrondir à la dizaine, centaine, au millier le plus proche Nb décimales doit être négatif Ex. : =ARRONDI(10927,25 ;-1)=10920 =ARRONDI(10927,25 ;-2)=10900 =ARRONDI(10927,25 ;-3)=11000
=AUJOURDHUI()	Donne la valeur de la date système (date du jour)
=MAINTENANT ( )	Donne la date et l'heure système, donc un nombre décimal.
=DATE ( année ; mois ; jour )	Donne le nombre correspondant à la date décrite. Ex : DATE(23;06;04) donne 8556
=JOUR (date)	Donne le jour de la date.
MOIS ( date )	Donne le mois de la date décrite.
ANNEE ( date )	Donne l'année de la date décrite.
=SI(Condition;Alors action si vrai;Sinon Action si faux)	Effectue une opération en fonction d'une condition posée, deux alternative : soit la condition est vraie, soit la condition est fausse.
=RECHERCHEV(Valeur recherchée;Table matrice;No index col;Valeur proche)	Permet de rechercher dans une table une valeur donnée dans la colonne située à l'extrême gauche d'un tableau et renvoie la valeur de la cellule spécifiée.

## V.3. Arrondir les calculs :

1. **Présentation** : Il est important de comprendre que le formatage d'une cellule est différent de l'arrondi d'un calcul.

**Le formatage d'une cellule** met en forme l'affichage de la valeur contenue dans la cellule. Ainsi la valeur **15,2665** formatée avec deux décimales sera affichée **15,27**.

Cependant, dans ses calculs, Excel **conserve toutes les décimales**, y compris celles qui ne sont pas affichées.

L'arrondi d'un calcul oblige Excel à **supprimer** les décimales excédentaires, pour ne retenir que le nombre de décimales spécifiées. Par exemple, du nombre **15,2665** Excel ne retiendra que la valeur **15,27** dans les opérations. Ce qui peut influencer sur le résultat des calculs, en particulier en matière comptable et commerciale.

## 2. Arrondir un calcul :

Pour arrondir le contenu d'une cellule, utilisez la syntaxe suivante :

**=ARRONDI (valeur ; nombre de décimales)**

Cette syntaxe arrondi le résultat à la valeur la plus proche. Ainsi :

**ARRONDI(3,156789 ; 2)** donne la valeur 3,16. (**2** chiffres après la virgule)

### Exemple 01 :

A	B
Formule	Description (résultat)
=ARRONDI(2,15 ;1)	Arrondit 2,15 à <b>une</b> décimale (2,2)
=ARRONDI(2,149;1)	Arrondit 2,149 à <b>une</b> décimale (2,1)
=ARRONDI(-1,475;2)	Arrondit -1,475 à <b>deux</b> décimales (-1,48)
=ARRONDI(1,4715;3)	Arrondit -1,4715 à <b>trois</b> décimales (1,472)
=ARRONDI(3,156789;0)	Arrondit 3,156789 à <b>zéro</b> décimale (3)
=ARRONDI(3,756789;0)	Arrondit 3,756789 à <b>zéro</b> décimale (4)
=ARRONDI(2 <b>1</b> ,5;-1)	Arrondit 21,5 à une décimale à gauche du séparateur décimal (20)
=ARRONDI(2 <b>5</b> ,5;-1)	Arrondit 25,5 à une décimale à gauche du séparateur décimal (30)

**3. Les variantes de la fonction ARRONDI:** La fonction ARRONDI propose les variantes suivantes :

**A. ARRONDI.INF :** La fonction ARRONDI.INF est similaire à la fonction ARRONDI, excepté qu'elle arrondit toujours le nombre en **tendant vers zéro**

Exemple	Résultat
=ARRONDI.INF(256,443; 2)	256,44
=ARRONDI.INF(256,447 ; 2)	256,44

**B. ARRONDI.SUP :** La fonction ARRONDI.SUP est similaire à la fonction ARRONDI, excepté qu'elle arrondit toujours le nombre en **s'éloignant de zéro**

Exemple	Résultat
=ARRONDI.SUP(256,443; 2)	256,45
=ARRONDI.SUP(256,447 ; 2)	256,45

**Exemple :**

L'exemple suivant vous montrera l'importance de la distinction entre le formatage d'une cellule et l'arrondi des calculs.

Saisissez le tableau ci-dessous sur une feuille de calcul Excel :

	A	B	C	D	E
1	Format deux décimales			Arrondi deux décimales	
2	PRIX HT	1289,56		PRIX HT	=B2
3	Taux de TVA	0,196		Taux de TVA	=B3
4	TVA	=B2*B3		TVA	=ARRONDI(E2*E3;2)
5	PRIX TTC	=B2+B4		PRIX TTC	=ARRONDI(E2+E4;2)
6	Taux de remise	0,05		Taux de remise	=B6
7	PRIX TTC NET	=B5-B5*B6		PRIX TTC NET	=ARRONDI(E5-E5*E6;2)

Vous devez obtenir les résultats suivants :

	B	C	D	E	F
1	<b>Format deux décimales</b>			<b>Arrondi deux décimales</b>	
2	PRIX HT	1 289,56 €		PRIX HT	1 289,56 €
3	Taux de TVA	19,60%		Taux de TVA	19,60%
4	TVA	252,75 €		TVA	252,75 €
5	PRIX TTC	1 542,31 €		PRIX TTC	1 542,31 €
6	Taux de remise	5,00%		Taux de remise	5,00%
7	PRIX TTC NET	1 465,20 €		PRIX TTC NET	1 465,19 €

L'écart peut sembler faible. Il faut cependant garder à l'esprit les points suivants :

- Plus les sommes impliquées sont importantes, plus les différences peuvent l'être également ;
- Arrondir les calculs répond à un impératif commercial : présenter des documents exacts ;
- Arrondir les calculs répond à un impératif fiscal, pour la TVA en particulier.

## VI.LA FORMULATION CONDITIONNELLE :

### V1.1. LA FONCTION SI() :

La fonction **SI** permet d'exécuter une opération si la condition est remplie et une autre si cette condition n'est pas remplie.

=SI(**Test**;Valeur1\_si\_VRAI; Valeur2\_si\_FAUX)

Alors

Sinon

Les conditions possibles (Test) sont :

<i>Supérieur</i>	>
<i>Supérieur ou égal</i>	>=
<i>Inférieur</i>	<
<i>Inférieur ou égale</i>	<=
<i>égal</i>	=
<i>différent</i>	<>

### Exemple 01 :

On dispose d'une liste d'étudiant de l'université de SBA et on désire vérifier s'ils appartiennent à la filière Informatique.

	A	B	C	D
1	Etudiant	Filière	Test	
2	Etudiant01	Electrique		
3	Etudiant02	Informatique		
4	Etudiant03	Economie		
5	Etudiant04	Informatique		
6	Etudiant05	Droit		
7	Etudiant06	Mecanique		
8	Etudiant07	Maths		
9	Etudiant08	Informatique		
10				

### Solution 01 :

	A	B	C	D	E
1	Etudiant	Filière	Test		
2	Etudiant01	Electrique	NON		
3	Etudiant02	Informatique	OUI		
4	Etudiant03	Economie	NON		
5	Etudiant04	Informatique	OUI		
6	Etudiant05	Droit	NON		
7	Etudiant06	Mecanique	NON		
8	Etudiant07	Maths	NON		
9	Etudiant08	Informatique	OUI		
10					


Formulaire en C2: `=SI(B2="Informatique";"OUI";"NON")`

Annotations:

- Test logique (pointe sur la cellule C2)
- Réponse si le test est **Vrai** (pointe sur "OUI")
- Réponse si le test est **Faux** (pointe sur "NON")

### Solution 02 :

Sélectionner la cellule C2 puis Allez dans le menu supérieur **Insertion** et cliquez sur **fonction** puis choisissez la fonction **SI**



Insérer une fonction

Recherchez une fonction :

Tapez une brève description de ce que vous voulez faire, puis cliquez sur OK.

Ou sélectionnez une catégorie : Les dernières utilisées

Sélectionnez une fonction :

- SI
- ARRONDI
- ET
- DATE
- MAINTENANT
- AUJOURDHUI
- ARRONDI.INF

**SI(test\_logique;valeur\_si\_vrai;valeur\_si\_faux)**

Vérifie si la condition est respectée et renvoie une valeur si le résultat d'une condition que vous avez spécifiée est VRAI, et une autre valeur si le résultat est FAUX.

[Aide sur cette fonction](#) [OK] [Annuler]

SI =SI(B2="Informatique";"OUI";"NON")

	A	B	C	D	E
	Etudiant	Filière	Test		
	Etudiant01	Electrique	"NON")		

*Test logique*

Arguments de la fonction

SI

Test\_logique B2="Informatique" = FAUX

Valeur\_si\_vrai "OUI" = "OUI"

Valeur\_si\_faux "NON" = "NON"

Vérifie si la condition est respectée et renvoie une valeur si le résultat d'une condition que vous avez spécifiée est VRAI, et une autre valeur si le résultat est FAUX.

Test\_logique est toute valeur ou expression et le résultat peut être VRAI ou FAUX.

Résultat = NON

[Aide sur cette fonction](#)

OK Annuler

*Réponse si le test est Faux*

*Réponse si le test est Vrai*

Valider par Ok et vous obtiendrez le même résultat que la solution 1

### Exemple 02 :

On dispose d'une liste d'étudiant et leurs moyennes générales semestrielles.  
On désire ajouter la mention admis si la moyenne générale est supérieure ou égale à 10  
ajourné avec une couleur de police rouge dans le contraire

	A	B	C
1	Etudiant	Moy Gle	Test
2	Etudiant01	14	
3	Etudiant02	12	
4	Etudiant03	8,5	
5	Etudiant04	5	
6	Etudiant05	15	
7	Etudiant06	13	
8	Etudiant07	9,5	
9	Etudiant08	9,25	
10			

et cas

	A	B	C
1	Etudiant	Moy Gle	Test
2	Etudiant01	14	Admis
3	Etudiant02	12	Admis
4	Etudiant03	8,5	Ajourné
5	Etudiant04	5	Ajourné
6	Etudiant05	15	Admis
7	Etudiant06	13	Admis
8	Etudiant07	9,25	Ajourné
9	Etudiant08	9,25	Ajourné
10			

=SI(B2>=10;"Admis";"Ajourné")

Pour la police rouge, on utilise la mise en forme conditionnelle (Voir 2<sup>ème</sup> cours page 26)

## VI.2. LA FONCTION SI AVEC 2 CONDITIONS

Les deux conditions doivent être reliées par une fonction logique :

**ET** lorsque les 2 conditions doivent être VRAIES

**OU** lorsque l'une des deux conditions doit être vérifiée

=SI(ET(Condition1;Condition2);Traitement si vrai;Traitement si faux)

=SI(OU(Condition1;Condition2);Traitement si vrai;Traitement si faux)

### Exemple :

Un étudiant est considéré comme "Excellent" s'il obtient 15 en anglais ET en espagnole, si non Pas excellent.

Dans ce cas de figure, il faut donc que l'étudiant ait 15 minimum dans chacune des 2 langues pour être considéré comme "Excellent".

	A	B	C	D
1	<b>Etudiant</b>	<b>Anglais</b>	<b>Espagnole</b>	<b>Mention</b>
2	Etudiant1	15	14	
3	Etudiant2	15	16	
4	Etudiant3	18	12	
5	Etudiant4	8	18	
6	Etudiant5	18	15	
7				

La formule en D2 est :

=SI(ET(B2>=15 ; C2>=15);" Excellent "; " Pas Excellent ")

	A	B	C	D
1	<b>Etudiant</b>	<b>Anglais</b>	<b>Espagnole</b>	<b>Mention</b>
2	Etudiant1	15	14	<b>Pas Excellent</b>
3	Etudiant2	15	16	<b>Excellent</b>
4	Etudiant3	18	12	<b>Pas Excellent</b>
5	Etudiant4	8	18	<b>Pas Excellent</b>
6	Etudiant5	18	15	<b>Excellent</b>
7				
8				

### VI.3.LA FONCTION SI() IMBRIQUÉE

Le traitement peut parfois dépendre d'une série de conditions imbriquées.  
On peut imbriquer jusqu'à 7 fonctions **SI()** les unes dans les autres.

**SI** condition1 vraie

**ALORS** traitement1

**SINONSI** condition2 vraie

**ALORS** traitement2

**SINON** traitement3

La **condition** peut être soit simple soit double (opérateur **ET**, **OU**).

Le nombre de conditions (**SI**) est inférieur d'une unité par rapport au nombre de **traitements** (pour 3 traitements, seulement 2 **SI**).

**Exemple 01 :**

On dispose d'une liste d'étudiant et leurs notes semestrielles.

Compléter la colonne Décision sachant que la décision dépend de la valeur de la moyenne générale:

<10 Ajourné(e)

>= 10 Admis(e)

>= 12 Admis(e) avec mention

	A	B	C	D	E	F
1	<b>Etudiant</b>	<b>Anglais</b>	<b>Maths</b>	<b>Informatique</b>	<b>MOY Gle</b>	<b>Décision</b>
2	Etudiant1	12	14	5	10,333	
3	Etudiant2	7	16	12	11,667	
4	Etudiant3	8	12	14	11,333	
5	Etudiant4	5	14	9	9,333	
6	Etudiant5	18	15	8,5	13,833	
7						

**SI** moyenne générale >= à 12

**ALORS** afficher Admis(e) avec mention

**SINONSI** moyenne générale >= à 10

**ALORS** afficher Admis(e)

**SINON** afficher Ajourné(e)

La formule en F2 sera donc :

=SI(E2>=12;" Admis(e) avec mention";SI(E2>=10;" Admis(e)";" Ajourné(e)))

Alors

Sinon Si

Alors

Sinon

	A	B	C	D	E	F
1	<b>Etudiant</b>	<b>Anglais</b>	<b>Maths</b>	<b>Informatique</b>	<b>MOY Gle</b>	<b>Décision</b>
2	Etudiant1	12	14	5	10,333	Admis(e)
3	Etudiant2	7	16	12	11,667	Admis(e)
4	Etudiant3	8	12	14	11,333	Admis(e)
5	Etudiant4	5	14	9	9,333	Ajourné(e)
6	Etudiant5	18	15	8,5	13,833	Admis(e) avec mention
7						

**Exemple02:**Interpréter l'exemple suivant :

=SI(A1="A";1;SI(A1="B";2;SI(A1="C";3;0)))

En littéraire ça donne

SI A1= A

--Si oui J'affiche la valeur 1

--Sinon je vérifie SI A1=B

----Si Oui j'affiche la valeur 2

----Sinon je vérifie SI A1=C

-----Si oui j'affiche la valeur 3

-----Sinon j'affiche la valeur 0

## VII.CALCUL CONDITIONNEL :

Les fonctions **NB.SI()** et **SOMME.SI()** effectuent un calcul uniquement sur les cellules correspondant à la condition spécifiée.

### 1. La Fonction NB.SI

#### NB.SI(plage; critère)

Cette fonction sert à déterminer le nombre de cellules non vides d'une plage qui répondent à un **critère** spécifié.

Si l'on souhaite dénombrer les valeurs supérieures à 0,5 dans la plage sélectionnée, on utilisera la fonction NB.SI :

=NB.SI(A1:D6;">0.5")

	A	B	C	D	E
1	0.3	0.9	0.5	0.4	
2	0.4	0.3	0.7	0.9	
3	0.1	0.1	0.3	0	
4	1	0.1	0.1	0.4	
5	0.3	0.5	0.2	0.8	
6	0	0.5	0.4	0.4	
7					
8	NB.SI	=NB.SI(A1:D6;">0.5")			

Il y a 5 valeurs strictement supérieures à 0,5 dans la plage sélectionnée.

	A	B	C	D	E
1	0.3	0.9	0.5	0.4	
2	0.4	0.3	0.7	0.9	
3	0.1	0.1	0.3	0	
4	1	0.1	0.1	0.4	
5	0.3	0.5	0.2	0.8	
6	0	0.5	0.4	0.4	
7					
8	NB.SI	5			

## 2. Fonction SOMME.SI()

Cette fonction vous permet d'additionner plusieurs cellules en fonction d'un critère déterminé.

Voici sa syntaxe : 
$$=SOMME.SI(plage;critère;somme\_plage)$$

- ✓ **plage** : est la plage de cellule contenant le critère à retenir
- ✓ **critère** : est LE critère à retenir (chiffre ou texte)
- ✓ **somme\_plage** : est la plage de cellule où se trouvent les nombres à additionner.

**Exemple01** : Dans la colonne A on a écrit des nombres. En colonne B, on a indiqué "garçon" ou "fille". On voudrait obtenir le total des nombres de la colonne A auxquels correspondent le mot "fille" en colonne B.

En A8 on écrit la formule suivante :

$$=SOMME.SI(B1:B7;"fille";A1:A7)$$

	A	B
1	28	Garçon
2	52	Fille
3	4	Garçon
4	55	Fille
5	1	Garçon
6	24	Garçon
7	54	Fille
8		

	A	B	C
1	28	Garçon	
2	52	Fille	
3	4	Garçon	
4	55	Fille	
5	1	Garçon	
6	24	Garçon	
7	54	Fille	
8	161		
9			

**Exemple 02:** Dans l'exemple suivant, saisir l'expression permettant de calculer le total des ventes réalisé par chaque vendeur.

1. Faite le calcule manuellement (formule)
2. Faite un calcule automatique (Insertion/Fonction)

	A	B	C	D
1	Vendeur	Montant		
2	Mohamed	100,000 د.ج.		
3	Salim	200,00 د.ج.		
4	Salim	300,00 د.ج.		
5	Mohamed	400,00 د.ج.		
6	Salim	700,00 د.ج.		
7	Salim	550,000 د.ج.		
8				
9	Total des ventes réaliser par Mohamed			
10	Total des ventes réaliser par Salim			
11				

**Solution 01 :**

En D9on écrit la formule suivante :=**SOMME.SI**(A2:A7;"Mohamed";B2:B7)

En D9 on écrit la formule suivante :=**SOMME.SI**(A2:A7;"Salim";B2:B7)

	A	B	C	D
1	Vendeur	Montant		
2	Mohamed	100,000 د.ج.		
3	Salim	200,00 د.ج.		
4	Salim	300,00 د.ج.		
5	Mohamed	400,00 د.ج.		
6	Salim	700,00 د.ج.		
7	Salim	550,000 د.ج.		
8				
9	Total des ventes réaliser par Mohamed			500
10	Total des ventes réaliser par Salim			1750
11				

**Solution 02 :** Sélectionnerla cellule D9puis Allez dans le menu supérieur **Insertion** et cliquez sur **fonction**puis choisissez la fonction**SOMME.SI**puis remplissez les cases comme suite :

	A	B	C	D	E	F	G
1	Vendeur	Montant					
2	Mohamed	100,000 د.ج.					
3	Salim	200,00 د.ج.					
4	Salim	300,00 د.ج.					
5	Mohamed	400,00 د.ج.					
6	Salim	700,00 د.ج.					
7	Salim	550,000 د.ج.					
8							
9	Total des ventes réaliser par Mo						
10	Total des ventes réaliser par S						
11							

Arguments de la fonction

SOMME.SI

Plage: A2:A7 = {"Mohamed";"Salim"}

Critère: A2 = "Mohamed"

Somme\_plage: B2:B7 = {100;200;300;400;}

= 500

Additionne des cellules spécifiées selon un certain critère.

Somme\_plage représente les cellules qui seront effectivement additionnées. Par défaut, les cellules dans la plage seront utilisées.

Résultat = 500

[Aide sur cette fonction](#) OK Annuler

Faite la même chose pour la cellule D10 (Salim) , la seul différence c'est qu'il remplir A3 qui correspond à Salim dans critère au lieu de A2

## VIII. AUTRES FONCTIONS

### 1. La Fonction NB :

**NB(valeur1;valeur2;...)**

La fonction **NB** permet de déterminer le total de cellules contenant des **nombre**s. Sélectionnez la fonction NB puis entrez simplement la/les cellules ou la/les plages qui serviront à déterminer le nombre total de cellules qui contiennent des nombres.

**Exemple** : Voici l'exemple suivant :

On veut compter le nombre de cellules qui contiennent des nombres

Inscrivez dans la cellule B7 la formule suivante :

=NB(A1 :B5)

	A	B
1	13	5
2	ABS	11
3		ABS
4	14	
5	8	15
6		
7		
8		

	A	B
1	13	5
2	ABS	11
3		ABS
4	14	
5	8	15
6		
7		5
8		

### 2. La Fonction NBVAL:

**NBVAL(valeur1;valeur2;...)**

Compte le nombre de cellules qui ne sont pas vides et les valeurs comprises dans la liste des arguments. Utilisez **NBVAL** pour compter le nombre de cellules contenant des données dans une plage ou une matrice.

Vous pouvez entrer jusqu'à 30 valeurs.

**Exemple** : Voici l'exemple suivant :

On veut compter le nombre de cellules qui contiennent des données (nombres et lettres)

Inscrivez dans la cellule B7 la formule suivante :

=NBVAL(A1 :B5)

	A	B
1	13	5
2	ABS	11
3		ABS
4	14	
5	8	15
6		
7		
8		

	A	B
1	13	5
2	ABS	11
3		ABS
4	14	
5	8	15
6		
7		8
8		

### 3. La fonction =NB.SI.ENS()

La fonction **NB.SI.ENS()** Compte le nombre de cellules non vides correspondant à plusieurs critères.

Voici sa **syntaxe** :

=NB.SI.ENS(**Plage\_critères1**;**Critères1**;**plage\_critères2**;**critèrese2**;....)

#### Exemple 01 :

Voici le tableau suivant:

Ecrire une fonction qui donne le **nombre d'étudiant** dont la **classe** et **CP** et le **sexe** et **Garçon**

=NB.SI.ENS(**B2:B13**;**CP**;**C2:C13**;**Garçons**)

=NB.SI.ENS(**B2:B13**;**B2**;**C2:C13**;**C2**)

	A	B	C	D
1	Nom	Classe	sexe	
2	MATA	CP	Garçon	
3	LOIPT	CE1	Garçon	
4	TROUNI	CE2	Fille	
5	NUIERA	CP	Garçon	
6	RATIN	CM1	Fille	
7	GARTIM	CM1	Fille	
8	BAARDI	CP	Garçon	
9	AGRAO	CE2	Fille	
10	ROIT	CP	Fille	
11	LIART	CM2	Garçon	
12	MASE	CM2	Fille	
13	GAREZ	CE1	Garçon	

### 4. La formule =SOMME.SI.ENS()

Cette formule permet d'additionner plusieurs cellules en fonction de plusieurs critères.

Voici sa **syntaxe** :

=SOMME.SI.ENS(**Plage\_somme**;**Plage\_critères1**;**Critères1**;**plage\_critères2**;**critères2**;....)

#### Exemple 01 :

Voici le tableau suivant:

Ecrire une fonction qui donne le **Nombre de produits** qui commencent par **A** et ont été **vendus** par **David**.

=SOMME.SI.ENS(**A2:A9**, **B2:B9**, "**=A\***", **C2:C9**, "**David**")

	A	B	C
1	Quantités vendues	Produit	Vendeur
2	5	Pommes	David
3	4	Pommes	Marie
4	15	Artichauts	David
5	3	Artichauts	Marie
6	22	Bananes	David
7	12	Bananes	Marie
8	10	Carottes	David
9	33	Carottes	Marie
10			

#### Exemple 02:

Nombre de produits qui ne sont pas des bananes et sont vendus par David.

=SOMME.SI.ENS(**A2:A9**, **B2:B9**, "**<>Bananes**", **C2:C9**, "**David**")

## 5. La formule =MOYENNE.SI.ENS

Cette formule renvoie la moyenne (arithmétique) de toutes les cellules qui répondent à plusieurs critères.

**Syntaxe:**

**=MOYENNE.SI.ENS(Plage\_moyenne;Plage\_critères1;Critères1;plage\_critères2;critères2;....)**

**Exemple 01:** Voici le tableau suivant:

Ecrire une fonction qui donne la Moyenne des notes comprises entre 70 et 90 au premier contrôle.

**=MOYENNE.SI.ENS(B2:B5; B2:B5;">70"; B2:B5;"<90")**

	A	B	C	D
1	Étudiant	Premier	Second	Final
2		Quiz	Quiz	Exam
3		Note	Note	Note
4	Emilio	75	85	87
5	Julie	94	80	88
6	Hans	86	93	Incomplet
7	Frédérique	Incomplet	75	75
8				

**Exemple 02:** Ecrire une fonction qui donne la Moyenne des notes à l'examen final supérieure à 80 pour tous les étudiants sans la note « Incomplet ».

**=MOYENNE.SI.ENS(D2:D5; D2:D5; "<>Incomplet"; D2:D5; ">80")**

## 6. La Fonction RECHERCHE :

**RECHERCHE(valeur\_recherchée; vecteur\_recherche; vecteur\_résultat)**

Cette fonction renvoie une valeur contenue dans un vecteur résultat (ligne ou colonne) associée à une valeur recherchée. (Les données recherchées doivent être triées en ordre croissant)

Prenons le tableau suivant, indiquant le classement à la coupe du monde de la FIFA (Internationale de Football Association) en 2010 :

Classement	Pays
1	Espagne
2	Pays-Bas
3	Brésil
4	Allemagne
5	Argentine
6	Uruguay
7	Angleterre
8	Portugal
9	Égypte
10	Chili

Pour retrouver le pays occupant une position donnée, on peut utiliser la fonction RECHERCHE :

On veut retrouver qu'elle est le pays qui occupe la première place

**=RECHERCHE(1;A2:A11;B2:B11)**

	A	B	C	D	E	F
1	Classement	Pays				
2	1	Espagne	RECHERCHE	=RECHERCHE(1;A2:A11;B2:B11)		
3	2	Pays-Bas				
4	3	Brésil				
5	4	Allemagne				
6	5	Argentine				
7	6	Uruguay				
8	7	Angleterre				
9	8	Portugal				
10	9	Égypte				
11	10	Chili				
12						

Ici, la **valeur\_recherchée** est «1», le **vecteur\_recherche** correspond au vecteur des positions et le **vecteur\_résultat** correspond au vecteur des pays.

	A	B	C	D	E	F
1	Classement	Pays				
2	1	Espagne	RECHERCHE	Espagne		
3	2	Pays-Bas				
4	3	Brésil				
5	4	Allemagne				
6	5	Argentine				
7	6	Uruguay				
8	7	Angleterre				
9	8	Portugal				
10	9	Égypte				
11	10	Chili				
12						

## 7. La Fonction RECHERCHEV :

**RECHERCHEV(valeur\_recherchée; matrice; no\_colonne)**

Cette fonction renvoie une valeur contenue dans une colonne résultat associée à une valeur\_recherchée d'une autre colonne.

### Exemple 01 :

Dans l'exemple suivant, on recherche la note qu'a obtenue l'élève portant le matricule 1071.

	A	B	C	D	E	F	G
1	Matricule	Note	Âge				
2	1736	80	19				
3	1549	63	22	RECHERCHEV	=RECHERCHEV(1071;A1:C10;2)		
4	1594	54	20				
5	1071	45	20				
6	1163	60	24				
7	1869	68	24				
8	1155	85	25				
9	1462	92	25				
10	1056	54	20				
11	...	...	...				
12							

=RECHERCHEV(1071;A1:C10;2)

La **valeur\_recherchée** est «1071», la **matrice** est la plage A1:C10 et le **no\_colonne** du résultat est 2 .

	A	B	C	D	E	F	G
1	Matricule	Note	Âge				
2	1736	80	19				
3	1549	63	22	RECHERCHEV	45		
4	1594	54	20				
5	1071	45	20				
6	1163	60	24				
7	1869	68	24				
8	1155	85	25				
9	1462	92	25				
10	1056	54	20				
11	...	...	...				
12							

On retrouve une note de 45

Pour retrouver l'âge de ce même étudiant, le no\_colonne doit être remplacé par la valeur 3.

=RECHERCHEV(1071;A1:C10;3)

### Exemple 02 : Notes scolaires

Voici un autre exemple avec une grille pour des notes scolaires.

Entrez le texte et les chiffres suivants dans les cellules appropriées.

Cet exercice consiste à afficher la bonne lettre pour la note dans la cellule B2 selon le chiffre situé dans la cellule B1.

	A	B
1	Note:	
2	Lettre:	
3		
4	0 e	
5	50 d	
6	60 c	
7	75 b	
8	85 a	

Dans la cellule B2, écrivez la formule suivante: =recherchev(b1;a4:b8;2)

#### 4.1. Le quatrième paramètre optionnel de Recherchev()

Pour rendre la fonction =recherchev ou =rechercheh encore plus puissante, il est aussi possible de comparer du texte. La première colonne du tableau de comparaison peut contenir des mots au lieu de chiffres. Comme pour les chiffres, les noms doivent être en ordre croissant. Contrairement aux chiffres, le texte doit être exact. Par exemple, le texte ne peut pas être entre Anjou et Avignon. Sinon, attendez-vous à des résultats bizarres comme vous verrez ci-dessous.

	A	B
1	Ville:	
2	Taux d'inoculation	=RECHERCHEV(B1;A4:B7;2)
3		
4	Anjou	0,17
5	Avignon	0,12
6	Berne	0,05
7	Bruxelle	0,03

- Entrez le texte, les chiffres et les formules suivantes dans les cellules appropriées.
- Dans la cellule B1, entrez Avignon.

Le résultat va afficher 12% dans la cellule B2. Mais il faut s'attendre à des résultats bizarres dans un cas.

- Dans la cellule B1, entrez Tokyo.

Le résultat contenu dans est de **3%**! Pourquoi? Parce qu'il est au-dessus de Bruxelles. Donc, il donne le contenu de la dernière ligne du tableau de comparaison.

C'est exactement pour cette situation que cette fonction offre un **quatrième paramètre optionnel**. Il vous permet de décider comment la fonction doit opérer. Doit-elle chercher **entre telle et telle valeur** ou chercher une valeur **exacte**?

Ce **quatrième paramètre** avec l'option **faux** ou **0** force la fonction à chercher la **valeur exacte**. Si le quatrième paramètre est égale à **vrai**, à **1** ou n'est pas présent, la fonction recherchev va chercher **entre tel ou tel valeur**. Donc, si la fonction aurait été écrite de cette façon =RECHERCHEV(B1;A4:B7;2;FAUX), il faudrait que le texte qui serait dans la cellule B1 soit écrit exactement de la même façon que l'une des valeurs du tableau de comparaison.

## 8. La Fonction RECHERCHEH :

**RECHERCHEH(valeur\_recherchée; matrice; no\_ligne)**

Cette fonction renvoie une valeur contenue dans une ligne résultat associée à une valeur\_recherchée d'une autre ligne.

- **Valeur\_recherchée** : Obligatoire. Représente la valeur à rechercher dans la première ligne de la table. Il peut s'agir d'une valeur, d'une référence ou d'une chaîne de texte.
- **matrice** : Obligatoire. Représente la table de données dans laquelle est exécutée la recherche de la valeur. Utilisez une référence à une plage ou un nom de plage.
- **no\_ligne** : Obligatoire. Représente le numéro de la ligne de la **matrice** à partir de laquelle la valeur correspondante est renvoyée.

Dans l'exemple suivant, on recherche le nombre de passagers qui sont embarqués à l'un des 6 premiers arrêts d'une ligne d'autobus (nombre de passagers qui ont embarqués dans le 1<sup>er</sup> arrêt)

	A	B	C	D	E	F	G
1	Arrêt	1	2	3	4	5	6
2	Nombre de passagers	8	9	6	2	11	5
3							
4	RECHERCHEH	=RECHERCHEH(1;A1:G2;2)					

**=RECHERCHEH(1;A1:G2;2)**

La **valeur\_recherchée** est «1» (premier arrêt), la **matrice** correspond à la plage A1:G2 et le **numéro\_ligne** du résultat est 2.

	A	B	C	D	E	F	G
1	Arrêt	1	2	3	4	5	6
2	Nombre de passagers	8	9	6	2	11	5
3							
4	RECHERCHEH	8					
5							

On retrouve une valeur de 8 passagers.

## 9. La Fonction ET :

**ET(test\_logique1; test\_logique2;...)**

Cette fonction renvoie la valeur «VRAI» si tous les tests\_logiques sont respectés et la valeur «FAUX» dans le cas contraire.

Prenons l'exemple suivant :

	A	B	C	D
1	4	ET	=ET(A1<20;A1>5)	
2	14			
3	24			

**ET(A1<20;A1>5)**

On effectue deux tests simultanément : on vérifie si la valeur de la cellule A1 est strictement inférieure à 20 et strictement supérieure à 5.

	A	B	C	D
1	4	ET	FAUX	
2	14			
3	24			
4				

10. **=Abs(cellule)** : Affiche la valeur absolue, ou positive, d'un chiffre.

Valeur	Abs(valeur)
25	25
-43	43

11. **=Ent(cellule)** : Affiche l'entier d'un chiffre. Elle élimine la fraction du chiffre.

Valeur	Ent(valeur)
25,99	25
43,55	43

12. **=fact(valeur) ou fact(cellule)** : Affiche le factoriel d'un nombre. Utile lors de calculs de probabilités. Ex.: =fact(3) = 6 (1 \* 2 \* 3)

11. **=mod(cellule1; cellule2)** : Calcule le reste de la division entière. Par exemple mod(13;4) = 1.

12. **=nb.vide(étendu)** : Affiche le nombre de cellules vides dans l'étendue choisie.

13. **=pair(valeur)** : Affiche la prochaine valeur paire si la valeur n'est pas paire.

Valeur	pair(valeur)
2	2
3	4

14. **=impair(valeur)** : Affiche la prochaine valeur impaire si la valeur n'est pas impaire.

Valeur	pair(valeur)
3	3
4	5

15. **=racine(valeur)** : Affiche la racine carrée d'un nombre ou d'une cellule. Ex.: =racine(9) = 3

16. **=ET()** et **=OU()** : La fonction **=Si()** peut être amélioré lorsqu'elle est utilisée avec d'autres fonctions telles que **=ET()** et **=OU()**. La partie qui suit démontre le fonctionnement de ces deux fonctions.

**=ET(condition1;condition2;condition3 ...)**

Il faut au moins deux conditions. Mais vous pouvez en ajouter d'autres. Toutes les conditions doivent être remplies pour que la fonction affiche VRAI. Sinon, elle affichera FAUX.

**=OU(condition1;condition2;condition3 ...)**

Il faut au moins deux conditions. Vous pouvez en ajouter d'autres. L'une des conditions doit être remplie pour que la fonction affiche VRAI. Si aucune des conditions n'est remplie, elle affichera FAUX.

Dans l'exemple suivant, les cellules **B1** et **B2** contiennent le texte "FAUX".

	A	B
1	0	=ET(A1=1;A2=1)
2	0	=OU(A1=1;A2=1)

### VIII. Chiffrés d'autres feuilles de calcul ou de classeur

Pour accéder à n'importe quelle cellule du classeur, il faut premièrement écrire le nom de la feuille de calcul, suivi du point d'exclamation et l'adresse de la cellule. Ex.: =Feuil1!A1. Il y a une autre manière d'arriver au même résultat.

=[test1.xls]Feuil1!\$A\$1 . Cela veut dire qu'il y a le nom du classeur entre crochets, suivi du nom de la feuille de calcul et du point d'exclamation et de l'adresse de la cellule.

### IX. Position absolue et relative:

Copier les formules est l'un des éléments qui accélèrent beaucoup la création de modèles. Il y a cependant un piège auquel il faut faire très attention. Prenons par exemple la formule suivante **A3=+A1+A2**. Si on recopie cette formule dans les cellules **B3** et **C3** le résultat serait le suivant: **B3=+B1+B2**, **C3=+C1+C2**. Pourquoi Excel n'a-t-il pas recopié exactement la formule **+A1+A2**?

A3		fx =A1+A2	
	A	B	C
1	10		
2	20		
3	30		
4			

B3		fx =B1+B2	
	A	B	C
1	10		
2	20		
3	30	0	0
4			

C3		fx =C1+C2	
	A	B	C
1	10		
2	20		
3	30	0	0
4			

Tout est dans la manière qu'Excel interprète la formule. Ce n'est pas l'adresse de la cellule qui est importante mais sa position par rapport à la cellule où se trouve la formule. Si le curseur est dans la **cellule A3**, Excel va interpréter la formule **+A1+A2** de cette manière suivante: la **cellule A1** est **deux cellules au-dessus** de la position actuelle du curseur et **A2** la **cellule juste au-dessus**. Donc, quand Excel va recopier la formule il va faire l'addition des deux cellules juste au-dessus de l'endroit où est écrite la formule. C'est pour cette raison que les formules recopiées en **B3** et **C3** le sont ainsi. On appelle ce fonctionnement des positions relatives. Relatif à l'emplacement de la formule, Excel va toujours chercher les données nécessaires au même endroit.

Si vous voulez conserver la référence de cellule d'origine dans cet exemple lorsque vous la copiez, vous devez utiliser une **référence de cellule absolue** en faisant précéder les colonnes (**A et B**) et la ligne (**2**) par un symbole dollar (**\$**). Ensuite, lorsque vous copiez la formule **=A\$2+B\$2** depuis la cellule **A3** vers **B3** et **C3**, la formule reste exactement identique.

A3					B3				
	A	B	C	D		A	B	C	D
1					1	10			
2	10				2	20			
3	30				3	30	30	30	
4					4				

C3				
	A	B	C	D
1	10			
2	20			
3	30	30	30	
4				

**Remarque 01 :** Dans des cas moins fréquents, vous pouvez utiliser une **référence de cellule mixte** en faisant précéder soit la **colonne**, soit la **ligne** par un symbole dollar afin de « verrouiller » soit la colonne, soit la ligne (par exemple, **\$A2** ou **B\$3**).

POUR UNE FORMULE COPIÉE :	SI LA RÉFÉRENCE EST :	ELLE PASSE À :
	\$A\$1 (colonne absolue et ligne absolue)	\$A\$1 (référence absolue)
	A\$1 (colonne relative et ligne absolue)	C\$1 (référence mixte)
	\$A1 (colonne absolue et ligne relative)	\$A3 (référence mixte)
	A1 (colonne relative et ligne relative)	C3 (référence relative)

**Remarque 02 :** On peut utiliser le signe "\$" pour "figer" ou "geler" une colonne ou une ligne dans une formule. Quand la formule sera recopiée, les parties gelées ne changeront pas.

- A1 Ni la colonne, ni la ligne ne sont gelées. Elles changeront avec la position de la formule.
- \$A1 La colonne est gelée mais pas la ligne.
- A\$1 La ligne est gelée mais pas la colonne.
- \$A\$1 La ligne et la colonne sont gelées.